Computer Drawn Isarithmic Maps

By Ole Hebin

Abstract

Description of a FORTRAN IV program which draws isarithmic maps in a matrix. All points on the isarithms are calculated by means of linear interpolation. If required, the maps can be provided with the values from the original matrix and with suitable headings.

Introduction

In the following a description is given of a FORTRAN IV program which draws isarithmic maps in a matrix with constant, but not necessarily the same distances between rows and columns.

All points on the isarithms are calculated by means of linear interpolation in triangles, the size of which depends on the input matrix and the chosen subprogram for plotting.

The values, or some of them, from the original matrix can be transferred to the isarithmic map. Furthermore, the user can freely choose a headline; this done, the equidistance will be drawn automatically.

The program is written for use on a Calcomp 563 Digital Plotter and has been tested at the Northern Europe University Computing Center (NEUCC), the Technical University of Denmark, Lundtofte, from where other computing centres are welcome to get a description of the applied plotter routines.

Description of the program

The plotter program is composed of a main program GRID0 and three subprograms GRID1, GRID2, and PLTTRI, all of them written in FORTRAN IV.

In order to facilitate the understanding, a schematic outline of the working process is given in fig. 1:

The main program GRID0 reads-in the parametres, the heading cards, the variable format cards, and the data matrix.

GRID0 draws map frames and heading and, dependent on the parametre GITT, it will call either GRID1 or GRID2 which again cals PLTTRI, where the actual drawing of the map will be carried



Fig. 1. Simplified flow-chart.

out. This completed, the program returns to GRID0 where the map may be provided with values from the original matrix A, the input parametres are printed out, and the program will stop.

PLTTRI

The central element in the map drawing program is the subroutine PLTTRI that draws isarithms in a triangle by connecting points of equal function value on the sides of the triangle according to the interpolation principle described below.

On fig 2 A and fig 2 B (x, y) indicates the coordinates of the point P and f the function value in (x, y).

If $f_1 \leqslant f_2 \leqslant f_3$; (1)

and h is the equidistance and

$$m = \text{integer}\left(\frac{f_3}{h}\right); \qquad (2)$$

$$n = \text{integer}\left(\frac{f_1}{h}\right);$$

then m–n isarithms will intersect the line P_1P_3.

These m–n isarithms will have the function values:

$$f_i = s \times h; \text{ for } s = n, n + 1, n + 2, \ldots, m \div n; \qquad (3)$$

and the coordinates:

$$\left.\begin{aligned} x_i &= \frac{x_3(f_1 \div s \times h) \div x_1(f_3 \div s \times h)}{f_1 \div f_3} \\ y_i &= \frac{y_3(f_1 \div s \times h) \div y_1(f_3 \div s \times h)}{f_1 \div f_3} \end{aligned}\right\} (4)$$

for $s = n, n + 1, n + 2, \ldots, m \div 2$



Fig. 2A. Triangle in which all interpolations are carried out.
Fig. 2B. Cross-section along P_1P_3 in figure 2A to show the positions of x_i, y_i and f_i.

If $f_2 \geqslant f_i$ the isarithm intersects the line $P_1 P_2$, if $f_2 \leqslant f_i$ the isarithm intersects the line P_2P_3.
The coordinates on either P_1P_2 or P_2P_3 are computed analog with (x_i, y_i, f_i), as (x_j, y_j, f_j).

When the points (x_i, y_i, f_i) and (x_j, y_j, f_j) have been computed, the line connecting them is drawn, and thus it continues till all m ÷ n isarithms have been drawn; this implies that according to condition (1) no isarithms will be missing.

If

$$f_1 \leqslant f_3 \leqslant f_2 \tag{1'}$$

(x_i, y_i, f_i) will be placed on P_1P_2 and

if

$$f_2 \leqslant f_1 \leqslant f_3 \tag{1''}$$

(x_i, y_i, f_i) will be placed on P_2P_3, etc.

This process is carried through for each call of the subroutine PLTTRI.

GRID1

The subroutine GRID1 computes on the basis of (xo, yo), Dx and Dy the position on the map of each A_{ij}. Furthermore, the mid-point (x_m, y_m) of each rectangle (Dx · Dy) is computed. (Fig. 3 A).

$$x_m = \frac{x_i + x_{i-1}}{2};\ y_m = \frac{y_i + y_{i-1}}{2};\ A_m = \frac{A_{i-1,\,j-1} + A_{i,j-1} + A_{i-1,j} + A_{i,j}}{4};$$

where i = 2, 3, 4, ..., N and j = 2, 3, 4, ..., M.

The subroutine PLTTRI is then called for each of the triangles 1, 2, 3, and 4.

This process will be repeated until the total matrix has been plotted out.

In order to speed up the plotting, decreasing and increasing i's are treated alternately. (Fig. 3 B).



Fig. 3A. Detail of grid-net when using GRID1.
Fig. 3B. Sequence of plotting in the grid to speed up the process.

GRID2

In principle, GRID2 works as GRID1 except that it makes a more detailed division of the rectangle (Dx · Dy) prior to the plotting.



Fig. 4. Detail of grid-net when using GRID2 showing the partition of each cell into 4 rectangles and each rectangle into 4 triangles.

First, Pa's coordinates are calculated as:

$$Y_a = \frac{Y_{j-1} + Y_j}{2};\ X_a = X_{i-1} \text{ and } f_a = \frac{A_{i-1,j-1} + A_{i-1,j}}{2}$$

second, Pb's coordinates as:

$$Y_b = Y_{j-1};\ X_b = \frac{X_{i-1} + X_i}{2} \text{ and } f_b = \frac{A_{i-1,j-1} + A_{i,j-1}}{2}$$

third, Pc's coordinates as:

$$Y_c = Y_a;\ X_c = X_b \text{ and } f_c = \frac{A_{i-1,j-1} + A_{i,j-1} + A_{i-1,j} + A_{i,j}}{4}$$

finally, Pm's coordinates as:

$$X_m = \frac{X_{i-1} + X_b}{2};\ Y_m = \frac{Y_{i-1} + Y_a}{2} \text{ and } f_m = \frac{A_{i-1,j-1} + f_a + f_b + f_c}{4}$$

In the resulting rectangle the plotting will be made as described for GRID1.

Next step will be a replacement of Pa by Pd and computation of a new Pm, whereafter the process will be repeated.

Then Pb is replaced by Pe and a new Pm computed. Finally, Pd will be replaced by Pa, and a last Pm is computed, which completes the plotting in the rectangle. In this way plotting continues until the whole matrix is drawn.

GRIDO

In GRIDO, parametres, heading, variable input-format, and data matrix must be read in before the scaling to desired size of map and drawing of map frames and heading can be carried out.

After call of either GRID1 or GRID2 all, or some, of the values from the original matrix may be transferred to the map.

The parametres are:

N	– number of columns in the matrix A (≤ 125)
M	– number of rows in the matrix A (≤ 125)
Xo	– the X-coordinate to the lower left corner of the map
Yo	– the Y-coordinate to the lower left corner of the map
Dx	– the distance between the columns in A
Dy	– the distance between the rows in A
H	– the equidistance
CM	– the size of the map (north-south) in cm. In case the read in map size should be less than 1 cm, the program will use 10 cm, and if the size should exceed 65 cm the program will use 65 cm as map size.

GITT – is GITT = 1, subroutine GRID1 will be used.
is GITT = 2, subroutine GRID2 will be used.

P – If P = 0, no plotting of the original matrix values will follow. For all other values of P, plotting will be effected.

RP – every RP^{th} row of A will be plotted

CP – every CP^{th} column of A will be plotted
If the total matrix A is requested, RP and CP must both of them be equal to 1.

INT – if INT = ÷ 1, the values of A will be plotted as integers, if INT > 0 the parametre INT indicates number of decimals in the plotted values.

The parametres are printed out as the last process in the program.

DECK SET-UP

Parametre cards.

Col.	Variable	Format	Examples	True value
1–3	N	I3	021	21
4–6	M	I3	018	18
7–11	Xo	F5.1	00000	0.0
12–16	Yo	F5.1	00000	0.0
17–21	DX	F5.1	00010	1.0
22–26	DY	F5.1	00010	1.0
27–31	H	F5.1	00500	50.0
32–34	CM	F3.1	250	25.0
35	GITT	I1	1	1
36	P	I1	1	1
37–38	RP	I2	02	2
39–40	CP	I2	02	2
41–42	INT	I2	÷1	÷1

Heading card

Column	text
1–25	heading
26–31	blank (here the program inserts the equidistance)
32–33	unit of the equidistance e.g. M., F., or Y.

Variable format card (only one card)

Column	text
1–80	format specification, F-type.



Fig. 5. Deck set-up.

Data

These are to be punched in accordance with the F-specification.

Running time

With GRID1, a 21×18 matrix with plotting of the total original matrix as integers runs for 2.26 min., and the plotting will take 32.8 min. When GRID2 is used, the running time will be 2.42 and 35.3 min. respectively. In both cases 506 lines are printed out by the line-printer.

Program listing with data examples:

```
$EXECUTE          IBJOB
$IBJOB
$IBFTC GRID0.
C
C     ...............................................................
C
C     MAIN PROGRAM  GRID0 (GIDH009)
C
C     PURPOSE
C        READ PARAMETERS AND DATA FOR CONSTRUCTION OF AND ISARITMIC MAP
C        BY CALLING SPECIFIED ROUTINES
C
C     USAGE
C        NORMAL
C
C     DESCRIPTION OF PARAMETERS
C        A    -  INPUT MATRIX (M*N)
C        N    -  NUMBER OF COLUMNS IN  A  (MAX. 125)
C        M    -  NUMBER OF ROWS IN  A  (MAX. 125)
C        X0   -  X-COORDINATE FOR LOWER LEFT CORNER OF MAP
C        Y0   -  Y-COORDINATE FOR LOWER LEFT CORNER OF MAP
C        DX   -  DISTANCE BETWEEN ORIGINAL COLUMNS
C        DY   -  DISTANCE BETWEEN ORIGINAL ROWS
C        H    -  CONTOUR INTERVAL
C        CM   -  HIGTH OF MAP IN CM (MAX. CM=65.)
C        GITT -  USE GRID1 (GITT=1), USE GRID2 (GITT=2)
C        P    -  POSTING DESIRED (P=1) ELSE (P=0)
C        CP   -  POSTING OF EACH CP'TH COLUMN
C        RP   -  POSTING OF EACH RP'TH ROW
C        INT  -  POSTING VALUES ARE PLOTTED AS INTEGERS (INT=-1), ELSE
C                AS PLOATING POINT NUMBERS (INT=NUMBER OF DECIMALS)
C        HEAD -  HEADING CARD (MAX. 33 CHARACTERS)
C        FORM -  INPUT FORMAT FOR A(M,N)
C
C     REMARKS
C        IF CM LESS THAN 1 - THE PROGRAM WIL USE CM=10, AND IF CM GREATE
C        R THAN 65 - THE PROGRAM WILL USE CM=65
C
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C        PARALF     -    TRANSLATES THE ORIGINAL COORDINATE SYSTEM TO A
C                        NEW ONE WITH A NEW ORIGIN BUT SAME DIRECTION OF
C                        AXIS
C        PLOT       -    DRAWS STRAIGHT LINES
C        FACTOR     -    SCALES ALL COORDINATE VALUES
C        SYMBOL     -    PLOTS SYMBOLS
C        NUMBER     -    PLOTS NUMBERS
C        GRID1      -    CALLS PLTTRI FOR DRAWING OF CONTOUR MAP
C        GRID2      -    CALLS PLTTRI FOR DRAWING OF CONTOUR MAP
C        PLTTRI     -    DRAWS CONTOURS OF A FUNCTION OF TWO VARIABLES
C                        INSIDE A SPECIFIED TRIANGLE
C
C     METHOD
C        REFER TO O. HEBIN,'COMPUTER DRAWN ISARITMIC MAPS', GEOGRAPHICAL
C        DEP., UNIVERSITY OF COPENHAGEN, 1969
C
C     ...............................................................
C
      REAL A(125,125),BUFFER(165)
      INTEGER HEAD(6 ), FORM(14), GITT, P, RP, CP
C
C
C     INPUT
C
      READ(5,101) N, M, X0, Y0, DX, DY, H, CM, GITT, P, RP, CP, INT
      READ(5,102) (HEAD(I),I=1,6 )
      READ(5,103) (FORM(I),I=1,14)
      DO 1 J=1,M
      I =M-J+1
    1 READ(5,FORM) (A(I,JA),JA=1,N)
C
C     SCALING
C
      IF(CM.LT.1.) CM = 10.
      IF(CM.GT.65.) CM = 65.
      FACTR=CM/FLOAT(M-1)
```

```
      DX=DX/DY
      DY=1.0
C
C     PLOTTING
C
      XM = X0+DX*FLOAT(N-1)
      YM = Y0+DY*FLOAT(M-1)
      YN=(YM+1.8)/2.0
      CALL PARALF(9.5,1.5)
      CALL FACTOR (FACTR)
      CALL PLOT (X0,Y0,2)
      CALL PLOT (X0,YM,1)
      CALL PLOT (XM,YM,1)
      CALL PLOT (XM,Y0,1)
      CALL PLOT (X0,Y0,1)
      CALL PLOT(X0-0.9,Y0-0.9,2)
      CALL PLOT(X0-0.9,YN,1)
      CALL PLOT(X0-0.9,YM+1.8,1)
      CALL PLOT(XM+0.9,YM+1.8,1)
      CALL PLOT(XM+0.9,YN,1)
      CALL PLOT(XM+0.9,Y0-0.9,1)
      CALL PLOT(X0-0.9,Y0-0.9,1)
      CALL SYMBOL (X0,YM+          0.5,0.5,HEAD(1),0.,33)
      CALL NUMBER (X0+10.57,YM+0.5,0.5,H,0.,-1)
      CALL SYMBOL (X0,Y0-0.5,0.25,19HHEBIN, GEOGR. DEP.  ,0.,19)
C
C     CHOOSE OF GRID AND PLOT CONTOURS
C
      IF(GITT.EQ.1) CALL GRID1 (N,M,X0,Y0,DX,DY,A,H)
      IF(GITT.EQ.2) CALL GRID2 (N,M,X0,Y0,DX,DY,A,H)
      IF(P.EQ.0) GO TO 3
C
C     PLOT OF ORIGINAL VALUES ON MAP IF REQUESTED
C
      DO 2 J=1,M,RP
      DO 2 I=1,N,CP
      X=  X0+DX*FLOAT(I-1)
      Y=  Y0+DY*FLOAT(J-1)
      CALL SYMBOL (X,Y,0.08,033,0.,-1)
      CALL NUMBER (X+0.1,Y,0.15,A(J,I),0.,INT)
    2 CONTINUE
    3 CONTINUE
C
C     DEPARTMENT OF PUBLIC RELATIONS
C
      WRITE(6,104)
      WRITE(6,105) HEAD
      WRITE(6,106) FORM
      WRITE(6,107) M, N, X0, Y0, DX, DY, CM, GITT
      WRITE(6,108) P, RP, CP, INT
      WRITE(6,109)
C
C     FORMATING
C
  101 FORMAT(2I3,5F5.1,F3.1,2I1,3I2)
  102 FORMAT(5A6,A3)
  103 FORMAT(13A6,A2)
  104 FORMAT(1H1,4X,76HUNIVERSITY OF COPENHAGEN, GEOGRAPHICAL DEP., OLE
     *HEBIN, GIOH009, APRIL 1969, /5X,51HPLOTTING OF ISARITMIC MAPS  ON
     *THE CALCOMP-PLOTTER.     )
  105 FORMAT(1H0/1H0,4X,8HHEADING,/1H0,4X,5A6,A3//)
  106 FORMAT(1H0,4X,13HINPUT FORMAT,/1H0,4X,13A6,A2//)
  107 FORMAT( 1H0,4X,21HSPECIFIED PARAMETERS,//
     *5X,79HNUMBER OF ROWS (M) ........................................
     *..................=,I10     /
     *5X,79HNUMBER OF COLUMNS (N) .....................................
     *..................=,I10     /
     *5X,79HX-COORDINATE LOWER LEFT CORNER (X0) .......................
     *..................=,F16.5   /
     *5X,79HY-COORDINATE LOWER LEFT CORNER (Y0) .......................
     *..................=,F16.5   /
     *5X,79HDISTANCE BETWEEN COLUMNS (DX) .............................
     *..................=,F16.5   /
     *5X,79HDISTANCE BETWEEN ROWS (DY) ................................
     *..................=,F16.5   /
```

```
     *5X,79HHIGTH OF PLOT IN CENTIMETERS (CM) .........................
     *.................=,F16.5    /
     *5X,79HWHAT TYPE OF GRID, (GITT) (GRID1=1, GRID2=2) ...............
     *.................=,I10      )
  108 FORMAT(
     *5X,79HIS POSTING DESIRED, (P) (NO=0, YES=1) .....................
     *.................=,I10      /
     *5X,79HEACH RP'TH ROW IS POSTED (RP) ............................
     *.................=,I10      /
     *5X,79HEACH CP'TH COLUMN IS POSTED (CP) ..........................
     *.................=,I10      /
     *5X,79HKIND OF POSTED NUMBER (INT) (INTEGER=-1, ELSE NUMBER OF DECI
     *MALS) ............=,I10      ///)
  109 FORMAT(1H0/1H0,4X,46H***** PLOTTING COMPLETED ***** I HOPE SO ****
     **)
      STOP
      END
$IBFTC GRID1.
C
C     ...............................................................
C
C     SUBROUTINE GRID1
C
C     PURPOSE
C        THIS SUBROUTINE PREPARES THE ORIGINAL DATAMATRIX FOR PLOTTING
C        BY THE ROUTINE PLTTRI, WHICH PERFORMS THE CONTOUR DRAWING
C
C     USAGE
C        CALL GRID1 (N, M, X0, Y0, DX, DY, A, H)
C
C     DESCRIPTION OF PARAMETERS
C        A     -  INPUT MATRIX (M*N)
C        N     -  NUMBER OF COLUMNS IN  A   (MAX. 125)
C        M     -  NUMBER OF ROWS IN  A   (MAX. 125)
C        X0    -  X-COORDINATE FOR LOWER LEFT CORNER OF MAP
C        Y0    -  Y-COORDINATE FOR LOWER LEFT CORNER OF MAP
C        DX    -  DISTANCE BETWEEN ORIGINAL COLUMNS
C        DY    -  DISTANCE BETWEEN ORIGINAL ROWS
C        H     -  CONTOUR INTERVAL
C
C     REMARKS
C        THIS ROUTINE HAS PARTLY BEEN CONSTRUCTED AND PROGRAMED IN ALGOL
C        BY THE DEPARTMENT FOR NUMERICAL ANALYSIS, MATHEMATICAL INSTITU
C        TE OF THE UNIVERSITY  OF COPENHAGEN. THE ROUTINE IS REWRITTEN
C        AND TRANSLATED TO FORTRAN IV FOR THE IBM7094 BY OLE HEBIN,
C        UNIVERSITY OF COPENHAGEN, GEOGRAPHICAL DEP., APRIL 1969
C
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C        THE ROUTINE PRESUPPOSES PLTTRI AND PLOT
C
C     METHOD
C        ALL GRIDPOINTS AND CONTOURPOINTS ARECALCULATED BY LINEAR INTER
C        POLATION.
C        REFER TO O. HEBIN,'COMPUTER DRAWN ISARITMIC MAPS', GEOGRAPHICAL
C        DEP., UNIVERSITY OF COPENHAGEN, 1969
C
C     ...............................................................
C
      SUBROUTINE GRID1 (N,M,X0,Y0,DX,DY,A,H)
      REAL  A(125,125),K
      K=0.5
      DO 1 J=2,M
      Y1=Y0+FLOAT(J-2)*DY
      Y2=Y1+DY
      YM=K*(Y1+Y2)
      DO 1 L=2,N
      I=L
      IF(MOD(J,2).EQ.0)I=N-L+2
      X1=X0+FLOAT(I-2)*DX
      X2=X1+DX
      XM=K*(X1+X2)
      F11=A(J-1,I-1)
      F12=A(J  ,I-1)
      F21=A(J-1,I  )
      F22=A(J  ,I  )
```

```
      FM=K**2*(F11+F12+F21+F22)
      CALL PLTTRI (XM,YM,FM,X1,Y1,F11,X2,Y1,F21,H)
      CALL PLTTRI (XM,YM,FM,X2,Y1,F21,X2,Y2,F22,H)
      CALL PLTTRI (XM,YM,FM,X2,Y2,F22,X1,Y2,F12,H)
    1 CALL PLTTRI (XM,YM,FM,X1,Y2,F12,X1,Y1,F11,H)
      RETURN
      END
$IBFTC GRID2.
C
C     ..............................................................
C
C     SUBROUTINE GRID2
C
C     PURPOSE
C        THIS SUBROUTINE CALCULATES BY LINEAR INTERPOLATION A GRID WITH
C        DOUBLE THE FINENESS OF THE ORIGINAL MATRIX AND PREPARES THE
C        DATA FOR PLOTTING BY THE ROUTINE PLTTRI WHICH PERFORMS THE CON
C        TOUR DRAWING
C
C     USAGE
C        CALL GRID2 (N, M, X0, Y0, DX, DY, A, H)
C
C     DESCRIPTION OF PARAMETERS
C        A     -  INPUT MATRIX (M*N)
C        N     -  NUMBER OF COLUMNS IN  A  (MAX. 125)
C        M     -  NUMBER OF ROWS IN  A  (MAX. 125)
C        X0    -  X-COORDINATE FOR LOWER LEFT CORNER OF MAP
C        Y0    -  Y-COORDINATE FOR LOWER LEFT CORNER OF MAP
C        DX    -  DISTANCE BETWEEN ORIGINAL COLUMNS
C        DY    -  DISTANCE BETWEEN ORIGINAL ROWS
C        H     -  CONTOUR INTERVAL
C
C     REMARKS
C        THIS ROUTINE HAS PARTLY BEEN CONSTRUCTED AND PROGRAMED IN ALGOL
C        BY THE DEP. FOR NUMERICAL ANALYSIS, MATHEMATICAL INSTITUTE OF
C        THE UNIVERSITY OF COPENHAGEN. THE ROUTINE IS REWRITTEN AND
C        TRANSLARED TO FORTRAN IV FOR THE IBM7094 BY OLE HEBIN, GEOGR.
C        DEP., UNIVERSITY OF COPENHAGEN
C
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C        THE ROUTINE PRESUPPOSES PLTTRI AND PLOT
C
C     METHOD
C        ALL GRIDPOINTS AND CONTOURPOINTS AR CALCULATED BY LINEAR INTER
C        POLATION.
C        REFER TO O. HEBIN,'COMPUTER DRAWN ISARITMIC MAPS', GEOGRAPHICAL
C        DEP., UNIVERSITY OF COPENHAGEN, 1969
C
C     ..............................................................
C
      SUBROUTINE  GRID2 (N,M,X0,Y0,DX,DY,A,H)
      REAL  A(125,125),K
      K=0.5
      DO 1  J=2,M
      Y1=Y0+FLOAT(J-2)*DY
      Y3=Y1+DY
      Y2=K*(Y1+Y3)
      DO 1 L=2,N
      I=L
      IF(MOD(J,2).EQ.0)I=N-L+2
      X1=X0+FLOAT(I-2)*DX
      X3=X1+DX
      X2=K*(X1+X3)
      F11=A(J-1,I-1)
      F13=A(J  ,I-1)
      F12=K*(F11+F13)
      F31=A(J-1,I  )
      F33=A(J  ,I  )
      F32=K*(F31+F33)
      F21=K*(F11+F31)
      F23=K*(F13+F33)
      F22=K**2*(F11+F13+F31+F33)
      XM=K*(X1+X2)
      YM=K*(Y2+Y3)
      FM=K**2*(F12+F13+F22+F23)
```

```
      CALL PLTTRI(XM,YM,FM,X1,Y2,F12,X2,Y2,F22,H)
      CALL PLTTRI(XM,YM,FM,X2,Y2,F22,X2,Y3,F23,H)
      CALL PLTTRI(XM,YM,FM,X2,Y3,F23,X1,Y3,F13,H)
      CALL PLTTRI(XM,YM,FM,X1,Y3,F13,X1,Y2,F12,H)
      XM=K*(X1+X2)
      YM=K*(Y1+Y2)
      FM=K**2*(F11+F12+F21+F22)
      CALL PLTTRI(XM,YM,FM,X1,Y1,F11,X2,Y1,F21,H)
      CALL PLTTRI(XM,YM,FM,X2,Y1,F21,X2,Y2,F22,H)
      CALL PLTTRI(XM,YM,FM,X2,Y2,F22,X1,Y2,F12,H)
      CALL PLTTRI(XM,YM,FM,X1,Y2,F12,X1,Y1,F11,H)
      XM=K*(X2+X3)
      YM=K*(Y1+Y2)
      FM=K**2*(F21+F22+F31+F32)
      CALL PLTTRI(XM,YM,FM,X2,Y1,F21,X3,Y1,F31,H)
      CALL PLTTRI(XM,YM,FM,X3,Y1,F31,X3,Y2,F32,H)
      CALL PLTTRI(XM,YM,FM,X3,Y2,F32,X2,Y2,F22,H)
      CALL PLTTRI(XM,YM,FM,X2,Y2,F22,X2,Y1,F21,H)
      XM=K*(X2+X3)
      YM=K*(Y2+Y3)
      FM=K**2*(F22+F23+F32+F33)
      CALL PLTTRI(XM,YM,FM,X2,Y2,F22,X3,Y2,F32,H)
      CALL PLTTRI(XM,YM,FM,X3,Y2,F32,X3,Y3,F33,H)
      CALL PLTTRI(XM,YM,FM,X3,Y3,F33,X2,Y3,F23,H)
    1 CALL PLTTRI(XM,YM,FM,X2,Y3,F23,X2,Y2,F22,H)
      RETURN
      END
$IBFTC PLTT.
C
C        ..............................................................
C
C        SUBROUTINE PLTTRI
C
C        PURPOSE
C           THIS SUBROUTINE DRAWS CONTOURS OF A FUNCTION OF TWO VARIAB-
C           LES INSIDE A SPECIFIED TRIANGLE BY CONNECTING POINTS ON THE
C           PERIMETER WITH THE SAME FUNCTION VALUE.
C
C        USAGE
C           CALL PLTTRI(X1,Y1,F1,X2,Y2,F2,X3,Y3,F3,H)
C
C        DESCRIPTION OF PARAMETERS
C           X1,Y1,F1,
C           X2,Y2,F2,
C           X3,Y3,F3, - ARE THE COORDINATES AND FUNCTION VALUES OF THE
C                       VERTICES OF THE TRIANGLE
C           H         - ARE THE VALUE DIFFERENCE BETWEEN CONTOURS
C
C        REMARKS
C
C
C           THIS SUBROUTINE HAS BEEN TRANSLATED FROM THE ALGOL PROCEDU
C           RE PLOTTRI SA 78IL BY O.HEBIN, GEOGRAPHICAL INSTITUTE, UNI
C           VERSITY OF COPENHAGEN. THE ORIGINAL PROGRAM WAS WRITTEN BY
C           BJ.SVEJGAARD AND P.LINDBLAD AT THE DEPARTMENT FOR NUMERICAL
C           ANALYSIS, MATHEMATICAL INSTITUTE OF THE UNIVERSITY OF COPE
C           NHAGEN.
C
C        SUBROUTINE AND FUNCTION SUBPROGRAMS REQUIRED
C           PLTTRI PRESUPPOSES SUBROUTINE  PLOT
C
C        METHOD
C           THE FUNCTIO VALUES ARE DETERMINED BY LINEAR INTERPOLATION
C           BETWEEN THE VERTICES
C
C        ..............................................................
C
      SUBROUTINE PLTTRI (X1,Y1,F1,X2,Y2,F2,X3,Y3,F3,H)
      XA=X1
      YA=Y1
      FA=F1
      XB=X2
      YB=Y2
      FB=F2
      XC=X3
```

```
      YC=Y3
      FC=F3
      IF(FA.LE.FB)GO TO 1
      U=XA
      XA=XB
      XB=U
      U=YA
      YA=YB
      YB=U
      U=FA
      FA=FB
      FB=U
    1 CONTINUE
      IF(FA.LE.FC)GO TO 2
      U=XA
      XA=XC
      XC=U
      U=YA
      YA=YC
      YC=U
       U=FA
      FA=FC
      FC=U
    2 CONTINUE
      IF(FB.LE.FC) GO TO 3
      U=XB
      XB=XC
      XC=U
      U=YB
      YB=YC
      YC=U
      U=FB
      FB=FC
      FC=U
    3 CONTINUE
      M=INT(FC/H)
      N=INT(FA/H)+1
      IF(M.LT.N) GO TO 10
      DO 9 I=N,M
      G=FLOAT(I)*H
      FA1=FA-G
      FB1=FB-G
      FC1=FC-G
      U=FA1-FC1
      P1=(XC*FA1-XA*FC1)/U
      P2=(YC*FA1-YA*FC1)/U
      IF(FB1.GE.0.)GO TO 4
      GO TO 5
    4 U=FA1-FB1
      Q1=(XB*FA1-XA*FB1)/U
      Q2=(YB*FA1-YA*FB1)/U
      GO TO 6
    5 U=FB1-FC1
      Q1=(XC*FB1-XB*FC1)/U
      Q2=(YC*FB1-YB*FC1)/U
    6 CALL PLOT(P1,P2,2)
    7 CALL PLOT(Q1,Q2,1)
    9 CONTINUE
   10 CONTINUE
      RETURN
      END
$DATA
 18 21      0      0  010   010 050030021 1 1-1
CONTOUR-MAP *** TEST ***        F.
(18F4.0)
 630 620 605 595 590 605 590 605 605 590 580 600 610 640 710105014001340
 640 605 580 560 610 595 598 590 600 600 575 610 620 86015001380 980 800
 640 600 650 570 570 600 580 605 570 580 590 585 85016001300 910 720 660
 800 690 640 600 570 600 585 605 570 630 580 800153014001210l030 595 598
 580 590 750 570 570 560 550 560 570 600 59513001370 995 900 750 580 590
 540 730 590 575 560 550 560 570 595 630110013801300 790 690 580 570 570
 550 610 490 500 530 600 570 560 605105012001050 850 850 590 570 580 610
 540 520 510 500 530 570 560 595 9001200 645 670 805 630 570 550 580 605
 500 495 540 540 550 565 580 60012501100 690 570 605 570 570 570 580 585
 475 525 520 500 520 560 580 7001250 850 700 580 550 505 490 550 700 550
 520 505 510 505 550 570 7001500 910 750 585 550 525 480 520 585 700 550
 500 530 500 520 605 590 950 900 650 585 500 505 500 470 510 520 500 480
 500 510 530 560 590 7901200 650 620 495 475 500 500 475 475 480 490 500
 505 510 550 595 6501350 950 640 475 480 475 500 480 510 520 460 510 460
 495 540 575 580 800 930 650 610 510 500 460 475 510 480 480 530 480 455
 500 480 560 700 850 595 510 495 485 500 500 470 500 460 460 470 510 480
 480 530 5901100 800 600 520 575 800 470 460 450 500 480 460 470 510 500
 485 485 6051000 700 570 540 495 500 490 475 470 440 420 420 445 465 460
 470 525 600 585 520 530 495 475 470 510 480 450 430 380 400 430 440 440
 580 580 800 570 505 515 500 510 500 480 505 420 400 400 430 440 450 460
 700 700 705 605 580 580 615 605 505 470 440 400 380 420 410 450 490 485
```

Examples:

The printed out parametres for the plotting process may look like this:

```
UNIVERSITY OF COPENHAGEN, GEOGRAPHICAL DEP., OLE HEBIN, GIOH009, APRIL 1969,
PLOTTING OF ISARITMIC MAPS  ON THE CALCOMP-PLOTTER.


HEADING,

CONTOUR-MAP *** TEST ***      F.


INPUT FORMAT,

(18F4.0)


SPECIFIED PARAMETERS,

NUMBER OF ROWS (M) ..........................................................=        21
NUMBER OF COLUMNS (N) .......................................................=        18
X-COORDINATE LOWER LEFT CORNER (X0) .........................................=         0.00000
Y-COORDINATE LOWER LEFT CORNER (Y0) .........................................=         0.00000
DISTANCE BETWEEN COLUMNS (DX) ...............................................=         1.00000
DISTANCE BETWEEN ROWS (DY) ..................................................=         1.00000
HIGTH OF PLOT IN CENTIMETERS (CM) ...........................................=        30.00000
WHAT TYPE OF GRID, (GITT) (GRID1=1, GRID2=2) ................................=         1
IS POSTING DESIRED, (P) (NO=0, YES=1) .......................................=         1
EACH RP'TH ROW IS POSTED (RP) ...............................................=         1
EACH CP'TH COLUMN IS POSTED (CP) ............................................=         1
KIND OF POSTED NUMBER (INT) (INTEGER=-1, ELSE NUMBER OF DECIMALS) ...........=        -1



***** PLOTTING COMPLETED ***** I HOPE SO *****

    PAPIRFORBRUG   40CM * 34CM        PLOTTETID  32.8 MIN        EFFEKTIVITET   42 PROCENT
```

On the back cover 3 maps (pl. 1, 2, and 3) drawn by the computer are inserted.

RESUMÉ

Nærværende artikel beskriver et FORTRAN IV program, der tegner isoliniekort i en matrice, hvor afstandene mellem rækker og søjler er konstante, men ikke nødvendigvis de samme for rækker og søjler.

Alle punkter på isolinierne beregnes ved lineær interpolation i trekanter, hvis størrelse afhænger af input-matricen og valg af subprogram for plotning.

Det udtegnede isoliniekort kan forsynes med originalmatricens talværdier eller dele heraf. Desuden kan kortet forsynes med en valgbar overskrift efter hvilken ækvidistancen automatisk udtegnes.

Programmet er skrevet for brug på en Calcomp 563 Digital Plotter og er afprøvet på Northern Europe University Computing Center [NEUCC], Danmarks Tekniske Højskole, Lundtofte, der vil kunne forsyne eventuelle brugere ved andre regnecentre med en beskrivelse af de anvendte plotterrutiner.